

di Francesco Petroni

Il Personal Computer IBM è ormai diventato lo standard nel campo delle macchine a 16 bit, così come a suo tempo l'Apple II stabilì di fatto uno standard per le macchine a 8 bit.

Questo sta comportando la proliferazione di macchine IBM compatibili ciascuna con un suo livello di compatibilità, e la proliferazione di SW intermedio e applicativo dedicato a questa famiglia di macchine che ne sfruttano al meglio le caratteristiche tecniche.

Per quanto riguarda il campo di cui ci interessiamo, la Computer Grafica, è questo forse l'elemento di minore "somiglianza" all'interno della famiglia, soprattutto per il fatto che le caratteristiche grafiche del PC IBM non sono molto sviluppate e offrire qualcosa di più in fondo è relativamente facile e questo può essere un elemento a favore del "concorrente" pur non precludendo la famigerata e tanto agognata "compatibilità".

Nei numeri scorsi abbiamo iniziato la trattazione di problemi di grafica con PC, affrontando la grafica fatta con "strumenti alfanumerici", ora affrontiamo la grafica fatta con tutti gli strumenti grafici hardware e software che l'IBM mette a disposizione.

## Computer Grafica su due monitor con il PC IBM

Per fare grafica con il Personal Computer IBM occorre disporre di una scheda grafica.

Tale scheda può essere sostitutiva, oppure aggiuntiva, della scheda interfaccia monocromatica/stampante; se è sostitutiva si perde la interfaccia stampante.

Questa incongruità ha scatenato i produttori indipendenti che producono schede multifunzione con uscite video alfanumeriche e/o grafiche, con uscite parallele e/o seriali, con espansioni RAM, ecc.

Il vantaggio di utilizzare la scheda grafica a colori IBM sta nel fatto che può lavorare contemporaneamente alla monocromatica; e lavorare con due monitor, e quelli IBM sono di ottima qualità, uno con il quale controllare programmi e dati e l'al-




Figura 1  
Programma  
Tavolozza  
Output.  
Nella singola  
casetta  
appaiono  
due colori  
combinati  
in percentuali  
del 75, 50, 25  
per cento.  
La diagonale  
rappresenta  
le combinazioni  
del colore  
con se stesso.

```

10 REM tavolozza di colori IBM
20 GOSUB 9100:CLS:COLOR 1,0
30 FOR V=0 TO 11:FOR D=0 TO 3:READ C1,C2
40 X1=0*79:Y1=V*16:LINE (X1,Y1)-(X1+30,Y1+14),,B
50 V1=V*2+1:V2=V1+1:O1=0*10+5
60 LOCATE V1,O1:PRINT C1;:LOCATE V2,O1:PRINT C2;
70 IF C1=0 AND C2=0 THEN GOTO 90
80 PAINT (X1+1,Y1+1),CHR$(C1)+CHR$(C2)
90 NEXT D:NEXT V
100 I$=INKEY$:IF I$="" THEN 100 ELSE GOSUB 9000:END
1000 DATA 0,0,0,17,0,34,0,51
1010 DATA 0,0,68,17,136,34,204,51
1020 DATA 0,0,85,17,170,34,255,51
1030 DATA 85,17,85,85,85,153,85,221
1040 DATA 68,17,85,85,102,153,119,221
1060 DATA 0,17,85,85,170,153,255,221
1070 DATA 170,34,170,102,170,170,170,237
1080 DATA 136,34,153,102,170,170,238,237
1090 DATA 0,34,85,102,170,170,255,237
1100 DATA 255,51,255,119,255,187,255,255
1110 DATA 204,51,221,119,238,187,255,255
1120 DATA 0,51,85,119,170,187,255,255
8900 GOSUB 9000:END
9000 REM switch to mono
9010 DEF SEG=0:POKE &H410,(PEEK(&H410) OR &H30)
9020 SCREEN 0:WIDTH 40:WIDTH 80
9030 RETURN
9100 REM switch to color
9110 DEF SEG=0:POKE &H410,(PEEK(&H410) AND &HCF) OR &H10
9120 SCREEN 0:WIDTH 40:SCREEN 1,0,0,0:RETURN

```

Figura 2 - Programma Tavolozza - Listato. I valori dei DATA da 0 a 255 sono ottenuti raggruppando 4 gruppi di 2 bit. Il gruppo 00 dà il colore 0, il gruppo 01 dà il colore 1, e così via.

tro, a colori, con il quale produrre le relative uscite grafiche, è molto divertente oltre che comodo.

La pagina video grafica è di 16K, e può lavorare in 4 modalità differenti variamente indirizzabili con le istruzioni SCREEN, COLOR, WIDTH:

SCREEN 2: una pagina schermo ad alta definizione 640 per 200 punti per 2 colori;

SCREEN 1: una pagina dello schermo a media definizione 320 per 200 punti per 4 colori;

SCREEN 0, 40 colonne: otto pagine di testo a 16 colori;

SCREEN 0, 80 colonne: quattro pagine di testo a 8 colori.

La combinazione delle istruzioni crea numerose situazioni di lavoro adatte ai vari problemi da affrontare. Per la trattazione di questo articolo tendente alla illustrazione e alla utilizzazione delle prestazioni tipicamente grafiche del PC IBM, impiegheremo solo la modalità media risoluzione.

La modalità grafica a media risoluzione permette, come detto, di lavorare su 320 per 200 punti, ciascuno dei quali indirizzabile individualmente e colorabile individualmente in uno dei 4 colori (in termini di pagina video: 320 per 200 per 2 diviso 8 uguale a 16K). Il colore di sfondo può essere scelto tra i 16 di base e gli altri tre dipendono dalla scelta di una tavolozza tra due disponibili.

Il linguaggio Basic, nella sua ultima versione (advanced Basic, release 2.0) contiene un poderoso set di istruzioni che permettono un facile sfruttamento delle capacità grafiche della macchina.

## Il passaggio dall'Apple II al PC IBM

Chi "nasce" graficamente in casa Apple ed ha sfruttato al massimo le sue prestazioni, trafficando con modulatori video e TV color utilizzando magari suggerimenti o apparecchiature (tipo tavoletta grafica) di MC, in definitiva risolvendo tutti i problemi classici della Computer Grafica (business graphic, bidimensionale, tridimensionale, animazione, game, ecc.) prova due sensazioni contrastanti, da una parte la soddisfazione di usare un SW potente e facile da maneggiare (che ha in più rispetto all'Apple le varie Window, Circle Put, Get, Draw, ecc.) e il piacere di vedere il risultato su di un monitor di ottima qualità, dall'altra parte la delusione nello scoprire che in fondo la definizione video non è stata spinta molto (solo 16K e una sola pagina per

```

100 REM Caricamento dati: * rem SICIUNO
110 GOSUB 9000:OPTION BASE 1:CLS:LOCATE 20,40:PRINT"attendere":NC=9
120 DIM X%(100,NC),Y%(100,NC),P$(NC),NS%(NC),CX(NC),YC(NC),YCZ(NC),DDZ(2002)
130 S=-.5:SX=7:SY=4:FOR I=1 TO NC:READ P$(I),NS%(I),CX(I),YCZ(I),YCZ(I)
140 XZ(I)=S*XZ(I)+SX:YZ(I)=S*YZ(I)+SY
150 FOR J=1 TO NSZ(I)+1:READ XZ(J,I),YZ(J,I)
160 XZ(J,I)=S*XZ(J,I)+SX:YZ(J,I)=S*YZ(J,I)+SY:NEXT J:NEXT I
200 REM Visualizzazione
210 GOSUB 9100:LINE (0,0)-(159,99),0,BF:LINE (160,0)-(319,99),2,BF
220 LINE (0,100)-(159,199),1,BF:LINE (160,99)-(319,199),0,BF
230 FOR I=1 TO NC:PSET (XZ(1,I),YZ(1,I))
240 FOR J=1 TO NSZ(I)+1:LINE -(XZ(J,I),YZ(J,I)),3:NEXT J
250 PAINT (XZ(I),YCZ(I)),CZ(I),3:NEXT I
260 GET (0,0)-(159,99),DDZ:PUT (160,0),DDZ,PRESET:PUT (0,100),DDZ,XOR
270 PUT (160,100),DDZ,OR:LINE (0,0)-(319,199),3,B:GOSUB 9000:END
9000 REM switch to mono
9010 DEF SEG=0:POKE &H410,(PEEK(&H410) OR &H30)
9020 SCREEN 0:WIDTH 40:WIDTH 80:RETURN
9100 REM switch to color
9110 DEF SEG=0:POKE &H410,(PEEK(&H410) AND &HCF) OR &H10
9120 SCREEN 0:WIDTH 40:SCREEN 1,0,0,0:RETURN
10000 DATA "Trapani",25,1,35,60
10010 DATA 26,89,19,79,9,74,5,63,10,56,7,52,11,42,15,34,22,32,28,24,33,25,40,36
10020 DATA 50,37,56,43,65,53,60,56,57,51,48,50,51,59,60,60,61,65,55,67,50,70,45
10030 DATA 77,46,86,26,89
11000 DATA "Palermo",50,3,100,65
11010 DATA 50,37,58,32,57,25,62,20,68,23,77,16,87,20,85,28,93,29,98,26,100,34
11020 DATA 116,41,134,32,149,36,149,36,158,45,157,54,160,61,156,69,148,73,145
11030 DATA 79,140,73,143,71,139,67,134,72,138,75,137,80,126,79,125,69,120,68
11040 DATA 113,72,102,78,87,77,88,88,82,89,82,83,83,77,77,81,73,80,72,74,60,71
11050 DATA 55,67,61,65,60,60,51,59,48,50,57,51,60,56,55,53,56,43,50,37
12000 DATA "Messina",30,2,200,40
12010 DATA 149,36,170,31,187,29,197,18,213,14,227,22,239,12,254,9,264,2,275,5
12020 DATA 269,8,267,12,267,17,265,19,262,27,253,40,244,55,230,46,218,47,214,41
12030 DATA 201,41,198,48,204,54,198,55,188,57,182,49,176,56,169,51,157,54
12040 DATA 158,45,149,36
13000 DATA "Agrigento",35,2,110,120
13010 DATA 137,134,123,131,114,128,111,123,103,117,95,116,85,111,69,96,57,94
13020 DATA 46,86,45,77,50,70,55,67,60,71,72,74,73,80,77,81,83,77,82,83,82,89
13030 DATA 88,88,87,77,102,78,113,72,122,73,122,79,119,81,116,84,111,83,113,95
13040 DATA 121,98,126,108,134,111,134,118,131,123,137,134
14000 DATA "Caltanissetta",33,1,130,100
14010 DATA 163,145,153,138,137,134,131,123,134,118,134,111,126,108,121,98
14020 DATA 113,95,111,83,116,83,119,81,122,79,122,73,113,72,120,68,125,69
14030 DATA 126,79,137,80,138,75,134,72,139,67,143,71,140,73,145,79,146,93
14040 DATA 139,100,142,109,155,106,167,115,167,125,175,128,172,140,163,145
15000 DATA "Enna",24,0,180,70
15010 DATA 157,54,169,51,176,56,182,49,188,57,198,55,195,66,203,78,202,88
15020 DATA 193,84,178,89,180,96,188,98,182,106,172,108,167,115,155,106,142,109
15030 DATA 139,100,146,93,145,79,148,73,156,69,160,61,157,54
16000 DATA "Catania",34,3,220,80
16010 DATA 244,55,238,69,236,82,228,92,230,99,229,105,214,100,200,103,202,109
16020 DATA 212,110,202,120,209,125,204,134,189,135,172,140,175,128,167,125
16030 DATA 167,115,172,108,182,106,188,98,180,96,178,89,193,84,202,88,203,78
16040 DATA 195,66,198,55,204,54,198,48,201,41,214,41,218,47,230,46,244,55
17000 DATA "Siracusa",26,1,220,140
17010 DATA 229,105,229,110,239,112,242,118,236,121,245,135,250,141,250,144
17020 DATA 239,149,233,157,231,167,235,180,230,183,223,178,222,170,210,165
17030 DATA 212,157,206,154,208,146,204,134,209,125,202,120,212,110,202,109
17040 DATA 200,103,214,100,229,105
18000 DATA "Ragusa",18,2,190,160
18010 DATA 223,178,218,178,218,179,212,174,205,177,197,176,187,171,175,169
18020 DATA 169,155,163,145,172,140,189,135,204,134,208,146,206,154,212,157
18030 DATA 210,165,220,170,223,178

```

Figura 3 - Programma SICIUNO - Listato. Lo scaling dei DATA viene fatto in sede di lettura per non appesantire la fase di visualizzazione. Il problema scaling si poteva, ovviamente, risolvere via SW di base, dimensionando lo schermo di lavoro con l'istruzione Window.

una macchina che indirizza 640 K) e che in fondo il SW grafico facilita il lavoro ma permette le stesse cose che, con uno sforzo sicuramente maggiore, si realizzano con le macchine 8 bit.

Tutti i "concorrenti" dell'IBM, non a caso, da una parte si dichiarano IBM compatibili, dall'altra tengono a offrire definizioni grafiche maggiori, più colori, ecc.

## I programmi

Presentiamo ora alcuni programmi.

Il primo si chiama Tavolozza (listato in fig. 2 e output in fig. 1) e serve per studiare tutte le combinazioni di due colori su quattro in un quadratino di due per due pixel.

Il secondo programma si chiama Sici-

liano (listato in fig. 3 e output in fig. 4) contiene i DATA relativi alla piantina della Sicilia e di cui realizza quattro uscite sulla stessa videata, impiegando le istruzioni PUT & GET.

Viene poi presentato il programma Siciliadue, che utilizza gli stessi DATA del precedente, ma lavora su due video (listato in fig. 6 e output in fig. 5). Sul video alfanumerico viene indicato lo scorrere delle province mentre l'immagine si forma sul video grafico.

Il quarto programma utilizza la istruzione DRAW per realizzare un set di caratteri utente. Tale set viene impiegato per indicare le province nella solita piantina della Sicilia. Si chiama Siciliatre, listato in figura 8 e output in figura 7.




Figura 4 - Programma SICIUNO - Output. La costruzione della figura (circa 250 dati), impiega circa 10 secondi. La istruzione PUT opera invece istantaneamente.

Infine un ultimo programma utilizza lo schermo alfanumerico in modalità TEXT, 80 colonne. In questa modalità, si può lavorare con 8 colori e 4 pagine (listato in fig. 10 e output, o meglio uno degli output, in fig. 9).

### L'istruzione Paint

È l'istruzione di riempimento di figura chiusa (in altri computer FILL); è molto efficace in quanto opera anche su figure molto frastagliate.

La sua sintassi prevede il passaggio di vari parametri, e cioè delle coordinate di un punto interno alla figura, del colore con il quale riempirla, del colore da considerare come confine del riempimento.

Il colore può essere un colore disponibile nella tavolozza dello SCREEN che si sta usando, oppure può essere costruito con la tecnica del retino.

In tal caso va definito un colore "utente" composto dall'accostamento di pixel di vario colore. Poiché il colore del singolo pixel è determinato da due bit, le posizioni che questo può assumere sono: 00, 01, 10, 11 e cioè 1, 2, 3, 4.

Per definire un retino composto da quattro punti del colore 2 e 3 va impostato un carattere, costituito da quattro gruppi di 2 bit, nel nostro esempio 2323 e cioè 10111011 e cioè 187.

Tale valore va passato come stringa CHR\$(187) nel campo previsto per il colore della sintassi del PAINT. Poiché tale colore creato dall'utente accetta la somma di stringhe, le combinazioni possibili sono infinite.

È chiaro che poiché la definizione dello SCREEN 1 è di 320 per 200 pixel, l'utiliz-

zo di colori intermedi ottenuti componendo pixel per pixel diminuisce tale definizione, ed è altrettanto chiaro che tale colore utente può essere usato per riempire aree ma non può esserlo per tracciare linee, e infine che il colore risultante non è compatto ma puntinato.

### La Tavolozza

Con il programma Tavolozza (figg. 1, 2) si sperimentano tutte le varie combinazioni di colore ottenute ponendosi due limiti: quello di combinare colori solo due a due e quello di combinarli solo nell'ambito di un quadratino due per due.

E poiché su quattro pixel ci sono 5 possibilità (4-0, 3-1, ecc.) con percentuali 100, 75, 50, 25, 0 di ogni colore rispetto all'altro, ed ogni colore può combinarsi con altri tre oltre che con se stesso, il totale dei casi è di 80.

Nel programma vengono eliminati i casi 4-0 e 0-4 e quindi ne rimangono 48. Poiché i 48 casi vengono presentati sotto forma di tabella, sono compresi i casi 3-1, 2-2, 1-3, di ciascun colore con se stesso. Sono infine compresi i doppietti dovuti alla simmetria della tabella.

In definitiva degli ottanta casi rimangono solo 22 di cui 4 colori puri, al 100 per 100.

### Il programma SICIUNO

Lavora sullo schermo 1 a quattro colori e realizza quattro piantine della Sicilia a quattro colori. Il colore definisce una provincia di cui sono noti i confini, memorizzati sotto forma di coordinate nella lista dei DATA.

I DATA contengono nome della provincia, numero dei punti che ne costituiscono il confine, numero del colore di riempimento e coordinata interna, poi le singole coordinate della linea chiusa che costituisce il confine.

I DATA sono in coordinate Apple II (279 per 191 pixel) e quindi subiscono in fase di lettura uno scaling. Subisce lo scaling anche la coordinata del punto interno.

Nei programmi seguenti useremo gli stessi DATA che quindi dovranno essere trasferiti.

Il Programma SICIUNO si svolge in due fasi, nella prima viene realizzata, nel quadrante in alto a sinistra del video, una piantina della Sicilia, questa viene poi memorizzata in un vettore opportunamente dimensionato, tramite l'istruzione GET.

Poi la piantina viene proiettata negli altri tre quadranti tramite la istruzione PUT. L'istruzione PUT accetta, oltre ai parametri che identificano la posizione della figura e il vettore in cui la figura viene memorizzata, anche un operatore logico che agisce a livello di pixel.

Ovvero i 2 bit identificanti il pixel della PUT sono combinati secondo l'operatore logico scelto con i bit del pixel sottostante.

Gli operatori logici sono AND, OR, XOR, PSET, PRESET; le combinazioni e quindi gli effetti infiniti.

### Programma SICIDUE

È il programma che utilizziamo per illustrare le modalità di lavoro su due schermi. Le routine per il passaggio sono la 9000, passaggio sullo schermo alfa, e la 9100 per il passaggio sullo schermo grafico.

Le routine sono quelle riportate sul manuale Basic, appendice I pagina 10.

Ricordiamo che i sistemi di computer grafica professionale prevedono in genere due schermi di lavoro, il che agevola moltissimo il controllo dei dati in fase di input, che è, come al solito, la fase più delicata. Tale controllo è essenziale ad esempio in CAD dove il DATO va esaminato numericamente e visivamente.

Pensiamo ad esempio ad un programma che visualizza un disegno identificato tramite le coordinate dei suoi punti (e questo è il metodo più diffuso). Solo un genio riuscirebbe ad immaginare cosa apparirà, solamente esaminando i dati numerici.

Le routine che permettono il passaggio tra i due schermi hanno l'inconveniente di cancellare quello su cui si passa.



Figura 5 - Programma SICIDUE - Output. Il metodo più semplice per rientrare sullo schermo grafico, conservandone il contenuto è quello di memorizzarlo in un array tramite la istruzione GET.

```

100 REM SICIDUE
110 GOSUB 9000:OPTION BASE 1:CLS:LOCATE 20,40:PRINT"attendere"
120 NC=9:DIM X%(100,NC),Y%(100,NC),P$(NC),NS%(NC),C$(NC),XC%(NC),YC%(NC)
130 DIM DD$(8008):S=1.05:SX=10:SY=2
140 FOR I=1 TO NC:READ P$(I),NS%(I),C$(I),XC%(I),YC%(I)
150 XC%(I)=S*XC%(I)+SX:YC%(I)=S*YC%(I)+SY
160 FOR J=1 TO NS%(I)+1:READ X%(J,I),Y%(J,I)
170 X%(J,I)=S*X%(J,I)+SX:Y%(J,I)=S*Y%(J,I)+SY:NEXT J:NEXT I
200 REM visualizzazione
210 CLS:GOSUB 9100:LINE (0,0)-(319,199),,B:GOSUB 9200:GOSUB 9000
220 FOR I=1 TO NC:LOCATE 18,40:PRINT "Piantina della Sicilia"
230 LOCATE 20,40:PRINT "Provincia di ";P$(I)
240 LOCATE 24,40:PRINT "Premi un tasto per continuare ";
250 I$=INKEY$:IF I$="" THEN 250
260 GOSUB 9100:GOSUB 9300:PSET (X%(I,I),Y%(I,I))
270 FOR J=1 TO NS%(I)+1:LINE -(X%(J,I),Y%(J,I)),3:NEXT J
280 PAINT (XC%(I),YC%(I)),C$(I),3:GOSUB 9200:GOSUB 9000:NEXT I
300 REM scritte
310 GOSUB 9100:GOSUB 9300:FOR I=1 TO NC:A%=XC%(I):B%=YC%(I)
320 LOCATE B%/8,A%/8-2:PRINT P$(I):NEXT I
330 GOSUB 9000:END
9000 REM switch to mono
9010 DEF SEG=0:POKE &H410,(PEEK(&H410) OR &H30)
9020 SCREEN 0:WIDTH 40:WIDTH B0:RETURN
9100 REM switch to color
9110 DEF SEG=0:POKE &H410,(PEEK(&H410) AND &HCF) OR &H10
9120 SCREEN 0:WIDTH 40:SCREEN 1,0,0,0:RETURN
9200 REM get
9210 GET (0,0)-(319,199),DD$:RETURN
9300 REM put
9310 PUT (0,0),DD$:RETURN
10000 REM data provincie

```

Figura 6 - Programma SICIDUE - Listato. Le subroutine 9000 e 9100 servono per il passaggio rispettivamente sul video alfanumerico e su quello grafico.

Quindi, a meno di trucchetti software, se si passa sullo schermo grafico dopo aver immesso, su quello alfanumerico, un dato, il soggetto va ridisegnato da capo aggiungendo il nuovo pezzo. Così pure se si passa

di nuovo sull'alfa per un nuovo input si perde la videata precedente.

Il sistema più semplice per rimediare a questo inconveniente è quello di parcheggiare in un vettore l'immagine precedente

tramite la GET grafica e di farla riapparire non appena tornati sullo schermo grafico.

È quello che facciamo nel programma SICIDUE, creando la visualizzazione di un messaggio sul video alfanumerico al momento del passaggio da una provincia all'altra.

Tralasciamo di descrivere il programma che è del tutto uguale al precedente.

In definitiva lavorare su due schermi contemporaneamente è particolarmente utile in molte situazioni, anche se fa venire ... il torcicollo.

È interessante sapere che anche i SW che vanno per la maggiore prevedono di lavorare con due video.

Il LOTUS 1 2 3 riconosce l'ambiente in cui lavora, quindi le sue potenti funzioni grafiche, che generano grafici (del tipo business) direttamente dai dati trattati internamente al worksheet, "escono" sul video grafico lasciando inalterato lo schermo alfa e tornandoci direttamente senza perdere l'immagine.

Poiché una volta definite le caratteristiche dei dati da graficizzare e in particolare le caselle in cui risiedono i valori numerici oggetto di analisi, queste vengono memorizzate, si possono ottenere aggiornamenti del grafico semplicemente premendo il tasto funzione 10, ovviamente dopo aver cambiato qualche numero.

Questo in accordo con il fatto che l'utente medio IBM non è l'hobbista, che comunque riesce a tirar fuori dalla sua macchina più di quanto questa "apparentemente" può dare, ma è il professionista o il dirigente che usa lo spreadsheet o il DataBase generator e che dei suoi dati vuole un'uscita grafica standard semplicemente premendo un tasto.

## Generazione di caratteri

Una delle necessità principali della Computer Grafica in tutte le sue applicazioni è quella di apporre scritte di opportuno formato e direzione sul disegno e nella posizione più opportuna. Sia la modalità SCREEN 2 (alta definizione) che la SCREEN 1 (media definizione) del PC IBM, permettono l'uso semplice della istruzione LOCATE Y, X e PRINT.

Nel primo caso le posizioni sono le 2000 del video 80 per 25 e nel secondo caso le 1000 del video 40 per 25, senza però possibilità di rotazioni né di ingrandimenti o rimpicciolimenti dei caratteri. Per quanto riguarda il set di caratteri disponibili è

identico a quello della modalità testo (SCREEN 0) per quanto riguarda i caratteri da 1 a 127.

Per quanto riguarda la parte superiore della tabella ASCII questa può essere definita dall'utente componendo i propri caratteri bit per bit e localizzandoli in una porzione della memoria lasciata libera.

La costruzione, poiché ogni pixel è definito da due bit, che ne specificano anche il colore, non è semplicissima, in quanto per definire un carattere in una matrice 8 per 8 occorrono 16 byte.

Se si vuole allineare il set di caratteri con le modalità testo, è disponibile sul dischetto del DOS 2.0 la tavola GRAFTABL (che occupa circa 1 Kbyte).

Un sistema per attivare, in maniera un po' pedestre, un proprio set di caratteri alternativi è quello di definirli in Basic tramite la istruzione DRAW. Questo sistema permette di utilizzare produttivamente alcune caratteristiche della istruzione DRAW e cioè l'orientamento (nelle quattro direzioni 0, 90, 180, 270 gradi), lo scaling e la possibilità di posizionare l'oggetto DRAW in una qualsiasi posizione, indipendentemente dal reticolo righe/colonne. C'è poi il vantaggio di poter definire il proprio set di caratteri in una matrice inferiore alla 8 per 8 e di poterli costruire con il criterio della proporzionalità. Questo permette di raggiungere, nello schermo 1, una definizione di circa 60 colonne.

La sintassi dell'istruzione è DRAW A\$, dove A\$ è una stringa di caratteri alfanumerici, ciascuno dei quali indica una direzione o una modalità di movimento, e numerici per indicare la quantità di movimento. Ad es. U3 vuol dire traccia verso l'alto (UP) un segmento dal punto dove stai, lungo tre pixel.

Nel programma SICITRE è inserito un set di caratteri (sono solo i 26 caratteri alfabetici maiuscoli) interni ad una matrice di 6 pixel di altezza e di 3÷6 pixel di larghezza. Con tale set si possono scrivere mediamente 60 caratteri per riga.

Una volta costruito un set il suo uso è particolarmente semplice, per ogni DRAW occorre definire la posizione (si può fare con la istruzione PSET) e tramite i parametri riconosciuti dal DRAW, colore, orientamento, scala.

Mentre la PRINT crea una finestra del colore background entro la quale scrive con il colore foreground, nel nostro caso dobbiamo evitare che il colore del DRAW coincida con quello dello sfondo, o altri-



Figura 7  
Programma  
SICITRE  
Output.  
Occorre  
stare attenti  
a non eseguire  
la DRAW  
con lo stesso  
colore  
dello sfondo.

```

100 REM SICITRE
110 GOSUB 1000:OPTION BASE 1:CLS:LOCATE 20,40:PRINT"attendere"
120 NN=26:DIM A$(NN):FOR I=1 TO NN:READ V$,A$(I):NEXT I
130 NC=9:DIM X$(100,NC),Y$(100,NC),P$(NC),NS$(NC),C$(NC),XC$(NC),YC$(NC)
140 S=1.05:SX=10:SY=2
150 FOR I=1 TO NC:READ P$(I),NS$(I),C$(I),XC$(I),YC$(I)
160 XC$(I)=S*XC$(I)+SX:YC$(I)=S*YC$(I)+SY
170 FOR J=1 TO NS$(I)+1:READ X$(J,I),Y$(J,I)
180 X$(J,I)=S*X$(J,I)+SX:Y$(J,I)=S*Y$(J,I)+SY:NEXT J:NEXT I
200 REM visualizzazione
210 CLS:GOSUB 1030:LINE (0,0)-(319,199),,B
220 FOR I=1 TO NC:PSET (X$(1,I),Y$(1,I))
230 FOR J=1 TO NS$(I)+1:LINE -(X$(J,I),Y$(J,I)),3:NEXT J
240 PAINT (XC$(I),YC$(I)),C$(I),3:NEXT I
300 REM scritte
310 FOR I=1 TO NC:PSET (XC$(I)-25,YC$(I)):FOR K=1 TO LEN(P$(I))
320 K#=MID$(P$(I),K,1):KK=ASC(K#)-64
330 C#="C"+STR$(C$(I)+3) MOD 4:DRAW C#+A$(KK):NEXT K:NEXT I
340 PSET (20,180):A#="SICILIA":FOR K=1 TO LEN(A#):K#=MID$(A#,K,1)
350 KK=ASC(K#)-64:DRAW "S"+A$(KK):NEXT K
360 GOSUB 1000:END
1000 REM switch to mono
1010 DEF SEG=0:POKE &H410,(PEEK(&H410) OR &H30)
1020 SCREEN 0:WIDTH 40:HEIGHT 80:RETURN
1030 REM switch to color
1040 DEF SEG=0:POKE &H410,(PEEK(&H410) AND &HCF) OR &H10
1050 SCREEN 0:WIDTH 40:SCREEN 1,0,0,0:RETURN
10000 REM data alfabeto
10010 DATA A,"U4 E2 F2 D4 U3 L3 BM+5,+3"
10020 DATA B,"U6 R3 F1 D1 G1 F1 D1 G1 L3 BR6"
10030 DATA C,"BR3 L2 H1 U4 E1 R2 BM+2,+6"
10040 DATA D,"U6 R3 F1 D4 G1 L3 BR6"
10050 DATA E,"U6 R3 L3 D3 R2 L2 D3 R3 BR2"
10060 DATA F,"U6 R3 L3 D3 R2 BM+3,+3"
10070 DATA G,"BR3 E1 U1 D1 G1 L2 H1 U4 E1 R2 BM+3,+6"
10080 DATA H,"U6 D3 R3 U3 D6 BR2"
10090 DATA I,"U6 BM+2,+6"
10100 DATA J,"BU6 R3 D5 G1 L1 H1 BM+5,+1"
10110 DATA K,"U6 D4 E4 G3 F3 BR2"
10120 DATA L,"BU6 D6 R3 BR2"
10130 DATA M,"U6 F2 E2 D6 BR2"
10140 DATA N,"U6 F4 U4 D6 BR2"
10150 DATA O,"BU1 U4 E1 R2 F1 D4 G1 L2 H1 BM+6,+1"
10160 DATA P,"U6 R3 F1 D1 G1 L2 BM+5,+3"
10170 DATA Q,"BU1 U4 E1 R2 F1 D4 G1 F1 H1 L2 BR5"
10180 DATA R,"U6 R3 F1 D2 G1 L1 F2 BR2"
10190 DATA S,"R3 E1 U1 H1 L2 H1 U1 E1 R2 BM+3,+6"
10200 DATA T,"BR2 U6 R2 L4 BM+6,+6"
10210 DATA U,"BU6 D5 F1 R2 E1 U5 BM+2,+6"
10220 DATA V,"BU6 D4 F2 E2 U4 BM+2,+6"
10230 DATA W,"BU6 D4 F2 E1 F1 E2 U4 BM+2,+6"
10240 DATA X,"U1 E4 U1 D1 G2 H2 U1 D1 F4 D1 BR1"
10250 DATA Y,"BR3 U3 E2 U1 D1 G2 H2 U1 BM+6,+6"
10260 DATA Z,"BR4 L4 U2 E4 L4 BM+2,+6"
20000 REM data provincie

```

Figura 8 - Programma SICITRE - Listato. I 26 DATA corrispondono ad un set di caratteri gestibili con la istruzione DRAW, che permette opzioni su SCALE, ROTATE, COLOR, ecc.




Figura 9 - Programma Prospetto - Output. Il passaggio tra le pagine è istantaneo e si ottiene cambiando il 3° e 4° parametro dell'istruzione SCREEN.

```

100 REM PROSPETTO DI UNA CASA
110 REM inizializzazioni
120 DIM C$(30),C%(3,7)
130 C$(1)=CHR$(205):C$(2)=CHR$(206):C$(3)=CHR$(196):C$(4)=CHR$(214)
140 C$(5)=CHR$(183):C$(6)=CHR$(204):C$(7)=CHR$(185):C$(8)=CHR$(186)
150 C$(9)=CHR$(177):C$(10)=CHR$(255):C$(11)=CHR$(203):C$(12)=CHR$(202)
160 C$(20)=STRING$(5,255):C$(21)=STRING$(72,205):C$(19)=STRING$(76,177)
200 REM
210 REM loop formazione 4 pagine
220 FOR P=0 TO 3:SCREEN 0,1,P,0
230 CLS:COLOR P+2,P,P+2
240 FOR O=5 TO 76:LOCATE 2,0:PRINT C$(3):NEXT O
250 LOCATE 2,4:PRINT C$(4):LOCATE 2,77:PRINT C$(5)
260 FOR O=3 TO 78:LOCATE 3,0:PRINT C$(1):NEXT O
270 LOCATE 3,4:PRINT C$(2):LOCATE 3,77:PRINT C$(2)
280 FOR V=4 TO 23:LOCATE V,4:PRINT C$(6):LOCATE V,77:PRINT C$(7):NEXT V
290 FOR V=3 TO 22:LOCATE V,5:PRINT C$(21):NEXT V:COLOR P+1,P,P+2
300 LOCATE 23,3:PRINT C$(19):LOCATE 12,3:PRINT C$(19):COLOR P+2,P,P+2
310 V1=4:O1=8:GOSUB 510:O1=27:GOSUB 510:O1=46:GOSUB 510:O1=65:GOSUB 510
320 V1=14:O1=8:GOSUB 510:O1=27:GOSUB 510:O1=46:GOSUB 510:O1=65:GOSUB 510:NEXT P
400 REM
410 PRINT CHR$(7):REM loop visualizzazione 4 pagine
420 FOR P=0 TO 3:SCREEN 0,1,P,P
430 I$=INKEY$:IF I$="" THEN 430
440 NEXT P:GOTO 420
500 REM
510 REM subroutine finestra
520 H=5:L=6:V3=V1+H:O3=O1+L
530 FOR V=V1 TO V3:LOCATE V,O1:PRINT C$(7):LOCATE V,O3:PRINT C$(6):NEXT V
540 LOCATE V1,O1:PRINT C$(11):LOCATE V1,O3:PRINT C$(11)
550 LOCATE V3,O1:PRINT C$(12):LOCATE V3,O3:PRINT C$(12):COLOR P+2,0,P+2
560 FOR V=V1+1 TO V3-1:LOCATE V,O1+1:PRINT C$(20):COLOR P+2,0,P+2:NEXT V
570 COLOR P+2,P,P+2:RETURN

```

Figura 10 - Programma Prospetto - Listato. Il programma utilizza lo SCREEN0, cioè quello TEXT. Se l'uscita è su monitor a colori si possono usare 8 colori su 4 pagine (in modalità 80 colonne).

menti dovremo creare anche noi una finestra.

Il formato del DRAW del carattere è proporzionale, parte dal basso a sinistra e finisce in basso a destra, quindi per scrivere

una stringa basta posizionare solo il primo carattere, e i successivi si posizionano automaticamente.

La routine di scrittura (riga 300-360) consiste nel riconoscere per ogni carattere

della stringa da stampare qual è il corrispondente elemento del vettore A\$(K) da tracciare. La A è il primo, la B il secondo e così via.

Se si rispetta nella tabella dei caratteri l'ordine ASCII, la traduzione si può fare tramite la istruzione ASC (riga 320).

## Il programma PROSPETTO

Il quinto programma lavora sullo schermo TEXT, sul monitor a colori, e utilizza tutte le quattro pagine disponibili.

Viene disegnato usando il set di caratteri esteso un prospetto di una casa in varie versioni cromatiche.

Il disegno viene realizzato sulle quattro pagine disponibili senza visualizzare la fase di disegno.

Terminata questa fase viene ottenuto il passaggio da uno schermo all'altro premeendo un tasto.

Il tutto è semplice, rimane il rimpianto di non poter usare direttamente il sistema di paginazione anche in modalità grafica. Praticamente l'istruzione che gestisce modalità e paginazione è la SCREEN che accetta come terzo e quarto parametro rispettivamente pagina visualizzata, pagina di lavoro.

Se queste due pagine coincidono si segue direttamente la formazione del disegno, se non coincidono, il disegno viene realizzato sulla pagina nascosta e, cambiando il parametro pagina visualizzata, quella nascosta appare istantaneamente.

Con l'Apple II si gestiscono 2 pagine grafiche, con l'IBM, modalità testo 40 colonne, le pagine sono 8.

La possibilità di eseguire paginazioni viene sfruttata nei programmi di animazione o di grafica interattiva.

Il programma è diviso in quattro parti:

- inizializzazione (riga 100-160) con la quale vengono definite le stringhe di caratteri grafici utilizzati per il disegno;

- creazione (riga 200-320) il loop sulla P gestisce pagina attiva e pagina visualizzata; anche il variare dei colori è stato legato alla variabile P, per semplificare il programma evitando l'uso di tabelle e variabili;

- visualizzazione (riga 400-440) altro loop sulla P, in cui pagina attiva e pagina visualizzata coincidono;

- finestra (riga 500-570) è la subroutine grafica che crea il disegno della finestra, il colore del "vuoto" della finestra è comunque il nero.

# LA FAMIGLIA DEI PERSONAL COMPUTER OLIVETTI



## FRIENDLY & COMPATIBLE

Anche in leasing con Olivetti Leasing

Questa famiglia di personal compatibili tra loro e con i più diffusi standard internazionali, non ha rivali per espandibilità e flessibilità. Prestazioni che su altri diventano opzionali, sui personal computer Olivetti sono di serie. Per esempio M24 offre uno schermo ad alta definizione grafica, ricco di 16 toni o di 16 colori e con una risoluzione di 600x400 pixel; mentre la sua unità base dispone di 7 slots di espansione, fatto questo che gli consente di accettare schede di espansione standard anche se utilizza un microprocessore a 16 bit reali (INTEL 8086). Ma ricchi vantaggi offrono anche tutti gli altri modelli.

Basti pensare che tutte le unità base includono sia l'interfaccia seriale che quella parallela. Oppure basti pensare all'ampia gamma di supporti magnetici: floppy da 360 a 720 KB o un'unità hard disk (incorporata o esterna) da 10 MB. La loro compatibilità, inoltre, fa sì che si possa far uso di una grande varietà di software disponibile sul mercato. Come, ad esempio, la libreria PCOS utilizzabile anche su M24. Come le librerie MS-DOS®, CP/M-86® e UCSD-P System®, utilizzabili sia da M20 che da M21 e M24.

# olivetti

MS-DOS è un marchio Microsoft Corporation  
CP/M-86 è un marchio Digital Research Inc.  
UCSD-P System è un marchio  
University of California, San Diego

Per maggiori informazioni inviare il coupon a Olivetti  
Divisione Personal Computer, Via Meravigli 12, 20123 Milano.

NOOME \_\_\_\_\_  
INDIRIZZO \_\_\_\_\_  
CITTA \_\_\_\_\_  
TELEFONO \_\_\_\_\_